

## Supplement B

SAS program to create imputation datasets

Lag and Lead variables indicate measures for those variables in the preceding (lag) month or in the following (lead) month.  
Mal1, Mal2 and Mal3 variables are indicators of afebrile or febrile parasitemia occurring 1, 2, or 3 months after the exposure.

```
/*Examine the number and proportion of missing values among variables of interest*/

*Categorical Variables;
proc freq data = diss.aim1_final2;
    tables sexa season                                     /*complete variables*/
        sector momed ses2 mompar hbasch                  /*time fixed covariates*/
        malaria mal1prev mal2prev laginfection           /*TV covariates*/
        nonmalanemia4                                     /*Exposures*/
        mal1 mal2 mal3 / missing;                         /*outcomes*/
run;

*continuous variables;
proc univariate data = diss.aim1_final2;
    var zheight zweight zwast lagzheight lag2zheight lagzweight
        lag2zweight lagzwast lag2zwast;
run;

*****;
proc mi data = diss.aim1_final2 nimpute = 50 out = mi_fcs seed=54321;
    /*list every categorical/binary variable in the class statement*/
    class childrnk2 lowh
        season sector hbasch
        malaria mal1 mal2 mal3 mal1prev mal2prev
        nonmalanemia4 lagnonmalanemia4 leadnonmalanemia4
        laginfection lag2infection infection
        momed ses2 mompar sexa;
```

```

/*var statement includes all variables that will be used in the imputation model*
 *in order of how you want them imputed (generally most complete to least complete) */

var sexa age_mo age_mosq age_mocu season                                     /*complete variables*/
      gestage childrnk2 lowh                                                 /*TF auxiliary variables*/
      sector momed ses2 mompar hbasch                                         /*time fixed covariates*/
      lagnonmalanemia4 leadnonmalanemia4 infection                         /*TV auxiliary variables*/
      lag2infection leadzheight leadzweight leadzwast
      malaria mal1prev mal2prev laginfection                                 /*TV covariates*/
      lagzheight lag2zheight lagzweight
      lag2zweight lagzwast lag2zwast

nonmalanemia4 zheight zweight zwast                                         /*Exposures*/

mal1 mal2 mal3;                                                               /*Outcomes*/

fcs plots = trace (mean std) nbiter = 40;          *request trace plots for continuous variables;
                                                *set number of iterations to 40 as convergence is an issue;

*Use regpmm (predictive mean matching) for continuous variables as convergence was an issue (determined by trace
plots) when using reg for cont. vars;

fcs regpmm (zheight= lagzheight lag2zheight leadzheight zweight season sector hbasch
            malaria mal1prev mal2prev mal1 mal2 mal3 nonmalanemia4 laginfection momed
            ses2 sexa mompar age_mo age_mosq age_mocu gestage childrnk2 lowh / details);

fcs regpmm (zweight= lagzweight lag2zweight leadzweight zheight
            season sector hbasch malaria mal1prev mal2prev mal1 mal2 mal3 nonmalanemia4
            laginfection
            momed ses2 sexa mompar age_mo age_mosq age_mocu gestage childrnk2 lowh /details);

fcs regpmm (zwast= lagzwast lag2zwast leadzwast season sector hbasch
            malaria mal1prev mal2prev mal1 mal2 mal3 nonmalanemia4 laginfection momed ses2 sexa
            mompar age_mo age_mosq age_mocu gestage childrnk2 lowh / details);

```

```

fcs regpmm (lagzheight = zheight lag2zheight leadzheight zweight season sector hbasch
            malaria mal1prev mal2prev mal1 mal2 mal3 nonmalanemia4 laginfection momed
            ses2 sexa mompar age_mo age_mosq age_mocu gestage childrnk2 lowh / details);

fcs regpmm (lag2zheight = zheight lagzheight leadzheight zweight season sector hbasch
            malaria mal1prev mal2prev mal1 mal2 mal3 nonmalanemia4 laginfection momed
            ses2 sexa mompar age_mo age_mosq age_mocu gestage childrnk2 lowh / details);

fcs regpmm (leadzheight = zheight lagzheight lag2zheight zweight season sector hbasch
            malaria mal1prev mal2prev mal1 mal2 mal3 nonmalanemia4 laginfection momed
            ses2 sexa mompar age_mo age_mosq age_mocu gestage childrnk2 lowh / details);

fcs regpmm (lagzweight= zweight lag2zweight leadzweight zheight
            season sector hbasch malaria mal1prev mal2prev mal1 mal2 mal3 nonmalanemia4
            laginfection
            momed ses2 sexa mompar age_mo age_mosq age_mocu gestage childrnk2 lowh /details);

fcs regpmm (lag2zweight= zweight lagzweight leadzweight zheight
            season sector hbasch malaria mal1prev mal2prev mal1 mal2 mal3 nonmalanemia4
            laginfection
            momed ses2 sexa mompar age_mo age_mosq age_mocu gestage childrnk2 lowh /details);

fcs regpmm (leadzweight= zweight lagzweight lag2zweight zheight
            season sector hbasch malaria mal1prev mal2prev mal1 mal2 mal3 nonmalanemia4
            laginfection
            momed ses2 sexa mompar age_mo age_mosq age_mocu gestage childrnk2 lowh /details);

fcs regpmm (lagzwast= zwast lag2zwast leadzwast season sector hbasch
            malaria mal1prev mal2prev mal1 mal2 mal3 nonmalanemia4 laginfection momed ses2 sexa
            mompar age_mo age_mosq age_mocu gestage childrnk2 lowh / details);

fcs regpmm (lag2zwast= zwast lagzwast leadzwast season sector hbasch
            malaria mal1prev mal2prev mal1 mal2 mal3 nonmalanemia4 laginfection momed ses2 sexa
            mompar age_mo age_mosq age_mocu gestage childrnk2 lowh / details);

fcs regpmm (leadzwast= zwast lagzwast lag2zwast season sector hbasch
            malaria mal1prev mal2prev mal1 mal2 mal3 nonmalanemia4 laginfection momed ses2 sexa
            mompar age_mo age_mosq age_mocu gestage childrnk2 lowh / details);

/*Use logistic for binary/ordinal variables, this assumes ordering of class variables if there are more than 2
levels*/
fcs logistic (mal1= mal2 mal3 mal1prev mal2prev malaria zheight zweight zwast lagzheight lagzweight

```

```

lagzwast lag2zheight lag2zweight lag2zwast laginfection season sector hbasch
nonmalanemia4 momed ses2 mompar sexa age_mo age_mosq age_mocu gestage childrnk2 /
descending details);

fcs logistic (mal2= mal1 mal3 mallprev mal2prev malaria zheight zweight zwast lagzheight lagzweight
lagzwast lag2zheight lag2zweight lag2zwast laginfection season sector hbasch
nonmalanemia4 momed ses2 mompar sexa age_mo age_mosq age_mocu gestage childrnk2 /
descending details);

fcs logistic (mal3= mal1 mal2 mallprev mal2prev malaria zheight zweight zwast lagzheight lagzweight
lagzwast lag2zheight lag2zweight lag2zwast laginfection season sector hbasch
nonmalanemia4 momed ses2 mompar sexa age_mo age_mosq age_mocu gestage childrnk2 /
descending details);

/*Use data augmentation method to make additional copies of dataset so enough of
   all cells are filled in. The data then get weighted based on the number of parameters in the model*/
fcs logistic (nonmalanemia4 = Mal1 mal2 mal3 malaria mallprev mal2prev zheight zweight zwast
lagzheight lagzweight lagzwast lag2zheight lag2zweight laginfection season
sector hbasch momed ses2 mompar sexa age_mo age_mosq age_mocu
gestage childrnk2 lagnonmalanemia4 leadnonmalanemia4 / descending details
LIKELIHOOD=AUGMENT);

fcs logistic (malaria= mal1 mal2 mal3 mallprev mal2prev zheight zweight zwast
lagzheight lagzweight lagzwast lag2zheight lag2zweight
laginfection season sector hbasch nonmalanemia4 momed ses2 mompar sexa age_mo
age_mosq age_mocu
gestage childrnk2 / descending details likelihood=augment);

fcs logistic (mallprev= mal1 mal2 mal3 mal2prev malaria zheight zweight zwast
lagzheight lagzweight lagzwast lag2zheight lag2zweight
laginfection season sector hbasch nonmalanemia4 momed ses2 mompar sexa age_mo
age_mosq age_mocu
gestage childrnk2 / descending details likelihood = augment);

fcs logistic (mal2prev= mal1 mal2 mal3 mallprev malaria zheight zweight zwast
lagzheight lagzweight lagzwast lag2zheight lag2zweight
laginfection season sector hbasch nonmalanemia4 momed ses2 mompar sexa age_mo
age_mosq age_mocu
gestage childrnk2 / descending details likelihood = augment);

/* use the discriminant function for binary/cat variables;
   classeffects= option allows bin/cat vars to be covariates*/

```

```
fcs discrim (hbasch infection laginfection lag2infection lowh momed ses2 mompar sector lagnonmalanemia4  
leadnonmalanemia4 / classeffects = include) ;  
run;  
  
*Make permanent dataset;  
data diss.aiml_mi_fcs_v3;  
  set mi_fcs;  
run;
```